NASA TECHNICAL NOTE

NASA TN D-2414

NASA TN D-2414

AN INITIAL VALUE METHOD FOR THE NUMERICAL TREATMENT OF THE ORR-SOMMERFELD EQUATION FOR THE CASE OF PLANE POISEUILLE FLOW

by Philip R. Nachtsheim

Lewis Research Center

Cleveland, Ohio

NATIONAL AERONAUTICS AND SPACE ADMINISTRATION • WASHINGTON, D. C. • AUGUST 1964

AN INITIAL VALUE METHOD FOR THE NUMERICAL TREATMENT OF THE ORR-SOMMERFELD EQUATION FOR THE CASE OF PLANE POISEUILLE FLOW

By Philip R. Nachtsheim

Lewis Research Center
Cleveland, Ohio

NATIONAL AERONAUTICS AND SPACE ADMINISTRATION

For sale by the Office of Technical Services, Department of Commerce,
Washington, D.C. 20230 -- Price $0.75

AN INITIAL VALUE METHOD FOR THE NUMERICAL TREATMENT OF THE ORR-SOMMERFELD EQUATION FOR THE CASE OF PLANE POISEUILLE FLOW

by Philip R. Nachtsheim

Lewis Research Center

SUMMARY

An exact numerical method is presented for the calculation of the eigenvalues in the problem of the stability of plane Poiseuille flow. The method appears to be rapid and highly accurate and can easily be generalized to solve more complex stability problems. The method of solution consists of treating the boundary value problem as an initial value problem. The results obtained agree closely with the numerical results of Thomas.

INTRODUCTION

The stability of plane Poiseuille flow has been studied by many authors. Considerable controversy has been generated by the contradictory conclusions reached. Heisenberg (ref. 1) concluded that plane Poiseuille flow became unstable at a sufficiently high Reynolds number, but he did not obtain a minimum critical Reynolds number. Subsequently, Lin (ref. 2) obtained a minimum critical Reynolds number of 5300 based on the maximum velocity in the center of the channel and its half-width. Both Heisenberg and Lin used asymptotic series. A different method was used by Pekeris (ref. 3), who concluded that the flow is stable at all Reynolds numbers. The disagreement between the results of Pekeris and those of Lin led von Neumann to suggest a direct numerical calculation. According to reference 4, calculations were performed in 1950 under the direction of von Neumann, Pekeris, and Lin by using a method devised by von Neumann. The results of these calculations were not published. In 1953, however, Thomas (ref. 4) published the results of his calculations, which indicate that plane Poiseuille flow becomes unstable at a Reynolds number of 5780. The direct numerical calculations made by Thomas were quite lengthy and, according to reference 5, the limited amount of work performed required 2 weeks of machine time on a high-speed electronic calculator.

A problem that besets the direct numerical integration of the disturbance equation is associated with the large values of the Reynolds number at which instability may be expected. The solution varies rapidly, and fine steps must be taken. Thomas, who used a finite-difference technique, overcame this difficulty and reduced the truncation error per step by introducing a new variable that is a discrete representation of the original stream function.

The present numerical method is based on step-by-step integration of the disturbance equation; hence, there is no need to introduce a variable defined only at discrete points. The present method, therefore, can be more easily generalized than the finite-difference methods to study stability problems of a more general nature than the stability of plane Poiseuille flow. A reduction in the truncation error per step is achieved by employing the special integration formula of Milne (ref. 6).

This report outlines the steps of the initial value technique (ref. 7) as applied to the problem of the stability of plane Poiseuille flow and determines a limited number of eigenvalues for the purpose of comparison with the results of Thomas.

FORMULATION OF THE PROBLEM

In the case of plane Poiseuille flow between parallel plates at $\overline{y} = 0$ and $\overline{y} = 2L$ with the velocity distribution $U = U_{max}\left[\frac{2\overline{y}}{L} - \left(\frac{\overline{y}}{L}\right)^2\right]$, the Orr-Sommerfeld equation is obtained from the first-order perturbation of the Navier-Stokes equation. The disturbance velocities are obtained from the stream function, which satisfies the continuity equation identically

$$\overline{\psi}(\overline{x},\overline{y},\overline{t}) = \overline{\varphi}(\overline{y})\exp[i\overline{\alpha}(\overline{x} - \overline{ct})] \tag{1}$$

from which

$$\overline{u} = \frac{\partial\overline{\psi}}{\partial\overline{y}} = \overline{\varphi}'(\overline{y})\exp[i\overline{\alpha}(\overline{x} - \overline{ct})] \tag{2}$$

and

$$\overline{v} = -\frac{\partial\overline{\psi}}{\partial\overline{x}} = -i\overline{\alpha\varphi}(\overline{y})\exp[i\overline{\alpha}(\overline{x} - \overline{ct})] \tag{3}$$

(Symbols are defined in appendix A.) The disturbance flow is taken to be periodic in the distance $\overline{x}$ in the direction of the flow. The positive quantity $\overline{\alpha}$ is the wave number of a disturbance wave and $\overline{c}_r$, the real part of $\overline{c}$, is the velocity of propagation of the wave. The imaginary part of $\overline{c}$ will determine whether the disturbance will grow ($\overline{c}_i > 0$) or decay ($\overline{c}_i < 0$) in time. The convenient complex notation is used herein. Physical meaning is attached only to the real part of disturbance quantities. Let $Re = U_{max}L/\nu$ denote the Reynolds number, and let dimensionless variables be introduced by replacing $\overline{y}$ by yL,

$\bar{x}$ by xL, $\bar{\alpha}$ by α/L, $\bar{t}$ by tL/U_{max}, $\bar{c}$ by cU_{max}, $\bar{\psi}$ by $LU_{max}\psi$, and $\bar{\varphi}$ by $LU_{max}\varphi$. The Orr-Sommerfeld equation is obtained by eliminating the pressure from the two momentum equations and has the following form for plane Poiseuille flow in terms of the dimensionless variables:

$$\varphi'''' - 2\alpha^2\varphi'' + \alpha^4\varphi = i\alpha Re\left[(2y - y^2 - c)(\varphi'' - \alpha^2\varphi) + 2\varphi\right] \tag{4}$$

Solutions of the differential equation for φ, for given α and Re, can be made to satisfy the boundary conditions that the disturbance velocities u and v (and, hence, φ and φ') vanish at the boundaries

$$y = 0 \qquad \varphi = \varphi' = 0 \tag{5}$$

$$y = 2 \qquad \varphi = \varphi' = 0 \tag{6}$$

only for the proper values (eigenvalues) of c. It is also of interest to determine the minimum critical Reynolds number, the lowest value of Re for which instability occurs.

Of primary interest with regard to equation (4) are the solutions that are even functions of y about the line $y = 1$. Since the velocity profile is an even function of y about the line $y = 1$, the disturbance can be separated into even and odd function parts. The former, which has a simpler flow pattern, usually gives a lower critical Reynolds number; hence, the second boundary condition (eq. (6)) at $y = 2$ is replaced by a condition at $y = 1$, namely,

$$y = 1 \qquad \varphi' = \varphi''' = 0 \tag{7}$$

INITIAL VALUE TECHNIQUE

The approach to the eigenvalue problem for fixed α and Re used herein is to find values of $c = c_r + ic_i$ (eigenvalues) for which equation (4) has solutions (eigenfunctions) that satisfy the boundary conditions.

Trial solutions are obtained by step-by-step numerical integration of the differential equation for the assumed initial values and an assumed value of c. The proper initial values and c are determined by an iterative process that selects the one solution that satisfies the boundary conditions.

Now equation (4) has four linearly independent solutions, some of which grow exponentially at a rapid rate. Hence, it is important to include as much information as possible about the wanted solution in the problem statement. The preceding is accomplished by starting at $y = 0$ with the proper boundary values and then integrating forward. Additional information is supplied by starting at $y = 1$ with the proper boundary values and then integrating backward. Next it is necessary to perform the process of matching in the middle. No attempt was made to find an optimum matching point; however, the choice $y_c = 0.5$ will tend to equalize the truncation error of the backward and forward solutions.

For computational purposes the solution is carried out in terms of the disturbance vorticity amplitude and the stream function amplitude. Instead of solving the fourth-order equation, a system of two second-order equations is solved, where s represents the disturbance vorticity amplitude

$$\varphi'' = s + \alpha^2\varphi \tag{8a}$$

$$s'' = \alpha^2 s + i\alpha Re\left[(2y - y^2 - c)s + 2\varphi\right] \tag{8b}$$

Equations (8) are solved subject to the boundary conditions (eqs. (5) and (7)).

For the forward solution the initial values at $y = 0$ are

$$\varphi_f = 0 \tag{9a}$$

$$\varphi_f' = 0 \tag{9b}$$

$$s_f = p \tag{9c}$$

$$s_f' = q \tag{9d}$$

The backward solution is started at $y = 1$ with

$$\varphi_b = 1 \tag{10a}$$

$$\varphi_b' = 0 \tag{10b}$$

$$s_b = r \tag{10c}$$

$$s_b' = 0 \tag{10d}$$

The condition $\varphi_b(1) = 1$ is a normalizing condition and fixes the size of the whole solution. Hence, in the forward solution the values p and q cannot be fixed arbitrarily but must be determined in the iterative process that attempts to match the solutions φ_f and φ_b at some common point y_c. The solution must be continuous, and matching requires at $y = y_c$ that

$$\varphi_f = \varphi_b \tag{11a}$$

$$\varphi_f' = \varphi_b' \tag{11b}$$

$$s_f = s_b \tag{11c}$$

$$s_f' = s_b' \tag{11d}$$

If these conditions are satisfied, all the higher derivatives agree, and the matching is accomplished.

The quantities $\varphi_f(y_c)$, $\varphi_f'(y_c)$, $s_f(y_c)$ and $s_f'(y_c)$ are functions of p, q, and c and the quantities $\varphi_b(y_c)$, $\varphi_b'(y_c)$, $s_b(y_c)$, and $s_b'(y_c)$ are functions of r and c. Successive changes are made in the first estimates of the parameters so that equations (11) are ultimately satisfied.

The Newton-Raphson method is used to fulfill the conditions imposed by equations (11). If the chosen values p, q, r, and c produce a solution that approximately satisfies equations (11), a better approximation is obtained by starting with $p + \Delta p$, $q + \Delta q$, $r + \Delta r$, and $c + \Delta c$ instead of p, q, r, and c. The quantities Δp, Δq, Δr, and Δc are solutions of the equations

$$\varphi_f - \varphi_b + \Delta p \frac{\partial}{\partial p}(\varphi_f - \varphi_b) + \Delta q \frac{\partial}{\partial q}(\varphi_f - \varphi_b) + \Delta r \frac{\partial}{\partial r}(\varphi_f - \varphi_b) + \Delta c \frac{\partial}{\partial c}(\varphi_f - \varphi_b) = 0 \tag{12a}$$

$$\varphi_f' - \varphi_b' + \Delta p \frac{\partial}{\partial p}(\varphi_f' - \varphi_b') + \Delta q \frac{\partial}{\partial q}(\varphi_f' - \varphi_b') + \Delta r \frac{\partial}{\partial r}(\varphi_f' - \varphi_b') + \Delta c \frac{\partial}{\partial c}(\varphi_f' - \varphi_b') = 0 \tag{12b}$$

$$s_f - s_b + \Delta p \frac{\partial}{\partial p}(s_f - s_b) + \Delta q \frac{\partial}{\partial q}(s_f - s_b) + \Delta r \frac{\partial}{\partial r}(s_f - s_b) + \Delta c \frac{\partial}{\partial c}(s_f - s_b) = 0 \tag{12c}$$

$$s_f' - s_b' + \Delta p \frac{\partial}{\partial p}(s_f' - s_b') + \Delta q \frac{\partial}{\partial q}(s_f' - s_b') + \Delta r \frac{\partial}{\partial r}(s_f' - s_b') + \Delta c \frac{\partial}{\partial c}(s_f' - s_b') = 0 \tag{12d}$$

in which the functions and the partial derivatives that constitute the coefficients are evaluated at y_c.

The partial derivatives are obtained by solving additional initial-value problems. These equations are obtained by partial differentiation of the terms in equations (8). The coefficients of equations (8) are analytic functions of y and the parameters α, Re, and c. The solutions of equations (8), therefore, have the same analytic properties and possess the required partial derivatives.

The quantities $\partial\varphi_f/\partial p \equiv \varphi_{f,p}$ and $\partial s_f/\partial p \equiv s_{f,p}$ for the forward solution satisfy the system of equations

$$\varphi''_{f,p} = s_{f,p} + \alpha^2\varphi_{f,p} \tag{13a}$$

$$s''_{f,p} = \alpha^2 s_{f,p} + i\alpha Re\left[(2y - y^2 - c)s_{f,p} + 2\varphi_{f,p}\right] \tag{13b}$$

With the initial conditions at $y = 0$

$$\varphi_{f,p} = 0 \tag{14a}$$

$$\varphi'_{f,p} = 0 \tag{14b}$$

$$s'_{f,p} = 0 \tag{14c}$$

$$s_{f,p} = 1 \tag{14d}$$

The quantities $\partial\varphi_f/\partial q \equiv \varphi_{f,q}$ and $\partial s_f/\partial q = s_{f,q}$ for the forward solution need not be computed by solving an initial value problem, but they can be obtained as a linear combination of the two previous solutions (φ_f, s_f) and $(\varphi_{f,p}, s_{f,p})$ since there are only two linearly independent solutions of the differential equation when $\varphi(0)$ and $\varphi'(0)$ are fixed at the value zero. Note that the differential equations (13) are the same as equations (8), and the only difference between the two sets of integrals (φ_f, s_f) and $(\varphi_{f,p}, s_{f,p})$ is the initial conditions satisfied by each set. The required integrals are given by

$$\varphi_{f,q} = \frac{\varphi_f - p\varphi_{f,p}}{q} \tag{15a}$$

$$\varphi'_{f,q} = \frac{\varphi'_f - p\varphi'_{f,p}}{q} \tag{15b}$$

and

$$s_{f,q} = \frac{s_f - ps_{f,p}}{q} \tag{16a}$$

$$s'_{f,q} = \frac{s'_f - ps'_{f,p}}{q} \tag{16b}$$

In particular

$$s_{f,q}(0) = 0$$

and

$$s'_{f,q}(0) = 1$$

The quantities $\partial\varphi_f/\partial c \equiv \varphi_{f,c}$ and $\partial s_f/\partial c \equiv s_{f,c}$ for the forward solution satisfy the system of equations

$$\varphi''_{f,c} = s_{f,c} + \alpha^2\varphi_{f,c} \tag{17a}$$

$$s''_{f,c} = \alpha^2 s_{f,c} + i\alpha Re\left[(2y - y^2 - c)s_{f,c} + 2\varphi_{f,c} - s_f\right] \tag{17b}$$

With the initial conditions at $y = 0$

$$\varphi_{f,c} = 0 \tag{18a}$$

$$\varphi'_{f,c} = 0 \tag{18b}$$

$$s_{f,c} = 0 \tag{18c}$$

$$s'_{f,c} = 0 \tag{18d}$$

For the backward solution the quantities $\partial\varphi_b/\partial r \equiv \varphi_{b,r}$ and $\partial s_b/\partial r \equiv s_{b,r}$ satisfy exactly the same system of equations (eqs. (13)) as do $\varphi_{f,p}$ and $s_{f,p}$ except that the initial conditions in this case are at $y = 1$

$$\varphi_{b,r} = 0 \tag{19a}$$

$$\varphi'_{b,r} = 0 \tag{19b}$$

$$s'_{b,r} = 0 \tag{19c}$$

$$s_{b,r} = 1 \tag{19d}$$

Finally, for the backward solution the quantities $\partial\varphi_b/\partial c \equiv \varphi_{b,c}$ and $\partial s_b/\partial c \equiv s_{b,c}$ satisfy exactly the same system of equations (eqs. (17)) as do $\varphi_{f,c}$ and $s_{f,c}$ except that the initial conditions at $y = 1$ in this case are

$$\varphi_{b,c} = 0 \tag{20a}$$

$$\varphi'_{b,c} = 0 \tag{20b}$$

$$s_{b,c} = 0 \tag{20c}$$

$$s'_{b,c} = 0 \tag{20d}$$

The quantities $\partial\varphi_b/\partial p$, $\partial\varphi_b/\partial q$, $\partial\varphi_f/\partial r$, $\partial\varphi'_b/\partial p$, $\partial\varphi'_b/\partial q$, $\partial\varphi'_f/\partial r$, $\partial s_b/\partial p$, $\partial s_b/\partial q$, $\partial s_f/\partial r$, $\partial s'_b/\partial p$, $\partial s'_b/\partial q$, and $\partial s'_f/\partial r$ are, of course, zero, since the variable in the numerator is independent of the variable in the denominator.

Equations (12) then reduce to the forms

$$\varphi_f - \varphi_b + \Delta p\ \varphi_{f,p} + \Delta q\ \frac{\varphi_f - p\varphi_{f,p}}{q} - \Delta r\ \varphi_{b,r} + \Delta c(\varphi_{f,c} - \varphi_{b,c}) = 0 \qquad (21a)$$

$$\varphi_f' - \varphi_b' + \Delta p\ \varphi_{f,p}' + \Delta q\ \frac{\varphi_f' - p\varphi_{f,p}'}{q} - \Delta r\ \varphi_{b,r}' + \Delta c(\varphi_{f,c}' - \varphi_{b,c}') = 0 \qquad (21b)$$

$$s_f - s_b + \Delta p\ s_{f,p} + \Delta q\ \frac{s_f - ps_{f,p}}{q} - \Delta r\ s_{b,r} + \Delta c(s_{f,c} - s_{b,c}) = 0 \qquad (21c)$$

$$s_f' - s_b' + \Delta p\ s_{f,p}' + \Delta q\ \frac{s_f' - ps_{f,p}'}{q} - \Delta r\ s_{b,r}' + \Delta c(s_{f,c}' - s_{b,c}') = 0 \qquad (21d)$$

Hence, there are four complex equations to determine the four complex quantities Δp, Δq, Δr, and Δc at each step of the iteration procedure.

Each step of the iteration scheme is carried out by starting with an estimate of p, q, r, and c and then integrating step-by-step the forward system of equations (eqs. (8)) with the initial conditions (eqs. (9)) together with the two perturbation systems of equations (eqs. (13) and (17)) with the initial conditions (eqs. (14) and (18)), respectively. Then the backward system is integrated (eqs. (8)) with the initial conditions (eqs. (10)) and the two perturbation systems, which are equations similar to equations (13) and (17) but with the initial conditions (eqs. (19) and (20), respectively). The forward and backward solutions are compared at the matching point, and the coefficients in equations (21) are evaluated. Equations (21) are then solved for Δp, Δq, Δr, and Δc, and this solution gives an estimate of the increments required for the next iteration.

Only variations with respect to the real parts of p, q, r, and c need be obtained by step-by-step integration. Since the solutions of equations (8) are analytic functions of p, q, r, and c, the real and imaginary parts of the complex derivatives appearing in the coefficients of equations (21) can be expressed in terms of derivatives with respect to real quantities only.

The differential equations written in real form are displayed in appendix B along with equations (21); appendix B indicates how the coefficients can be written in terms of derivatives with respect to real quantities only.

The labor of carrying out the step-by-step integration can be reduced by the use of special formulas for integrating second-order differential equations in which the first derivative does not appear explicitly. In addition the truncation error per step is reduced by the use of such formulas. These integration formulas evaluate the second derivative at each step. Thus, corresponding to equations (8) written in real form there will be four second-derivative evaluations required; also, corresponding to the two perturbation systems of equations, equations (13) and (17) written in real form, there will be eight additional second-derivative evaluations required. Hence, in advancing the solution there are 12 second-derivatives to be evaluated at each step. The special integration formulas for starting and advancing the solution are given in appendix C.

RESULTS AND COMPARISONS

The procedure outlined previously for finding the eigenvalue c for a given point in the α,Re-plane was programed for solution by using double-precision arithmetic (16 significant figures) on the IBM 7094 computer located at the Lewis Research Center. A brief description of the program is given in appendix D, and a listing of the program is given in appendix E. The forward solutions (started at $y = 0$) were matched with the backward solutions (started at $y = 1$) at $y = 0.5$. Eigenvalues were calculated at a limited number of points in the α,Re-diagram, namely, at $\alpha = 1$ and Re = 1600, 2500, 6400, and 10,000 in order to compare the results of the present method with the results of Thomas.

Before making the comparison, however, it is appropriate to examine the accuracy of the present results and to consider the rate of convergence of the iterative process that determined the eigenvalues.

The accuracy of the results was examined at the point $\alpha = 1$ and Re = 10,000. Since the truncation error per step involved in integrating the differential equations increases as αRe increases, the results for lower values of Re should be more accurate than those at Re = 10,000. The accuracy of the results at this point was established by examining the eigenvalues and eigenfunctions when the example was rerun at a reduced step size. When the original solution, obtained for 128 steps, was rerun at 256 steps, the eigenvalues and the eigenfunctions obtained agreed to within four decimal places. This agreement indicates that the results are accurate to at least four decimal places.

An idea concerning the rate of convergence to an eigenvalue can be formulated from table I where the history of the various iterations is displayed. The eigenvalues at $\alpha = 1$ and Re = 2500 were being sought, and the eigenvalues and the initial values at $\alpha = 1$ and Re = 1600 were used as initial estimates. Iteration was stopped when

TABLE I. - HISTORY OF CONVERGENCE FOR WAVE NUMBER OF 1,

REYNOLDS NUMBER OF 2500, 128 STEPS

c_r	c_i	p_r	p_i	q_r	q_i	r_r	r_i
0.3231	-0.0262	19.8219	-11.4855	-31.0653	507.6370	-2.9503	0.1142
.3231	-.0280	28.1241	-25.5771	82.9764	1062.0009	-2.9495	.1220
.2886	-.0352	14.3404	-20.3758	240.0751	636.0073	-2.7970	.1407
.2879	-.0348	20.2816	-16.8579	94.8133	707.6264	-2.8018	.1368
.2919	-.0295	20.8031	-17.3995	64.3360	737.6275	-2.8196	.1177
.2973	-.0203	23.9439	-18.1969	12.9200	808.5010	-2.8440	.0829
.2979	-.0186	24.1864	-17.8535	-2.9052	808.4881	-2.8466	.0754
.3013	-.0147	25.2129	-18.1846	-22.8133	835.5380	-2.8611	.0603
.3011	-.0144	25.2429	-18.0949	-25.5464	834.2570	-2.8604	.0590
.3011	-.0142	25.2891	-18.0835	-26.9706	834.8310	-2.8607	.0580
.3012	-.0142	25.2889	-18.0832	-26.9715	834.8209	-2.8607	.0580

all the values for two consecutive iterations agreed to four decimal places. Similar runs were made to obtain other eigenvalues; for example, the eigenvalues at $\alpha = 1$ and Re = 6400 were obtained by using the eigenvalues and initial values at $\alpha = 1$ and Re = 2500. Convergence in this case required 18 iterations to achieve four-decimal-place agreement between two consecutive runs. About 25 iterations can be performed in 1 minute.

Reynolds number, Re	Method of Thomas		Present method (128 steps)	
	c_r	c_i	c_r	c_i
1,600	0.3231	-0.0262	0.3231	-0.0262
2,500	.3011	-.0142	.3012	-.0142
6,400	.2569	.0009	.2569	.0010
10,000	.2375	.0037	.2375	.0038

The table at the left shows the results obtained by using the present method and the results obtained by Thomas (ref. 4) for $\alpha = 1$ and various Reynolds numbers. As can be seen from the table, the results differ at most by one unit in the fourth decimal place.

Table II shows the eigenfunctions at $\alpha = 1$ for Re = 10,000 for a 256-step solution. The results presented in this table can be compared with results given in table V of reference 4. In making this comparison, it must be remembered that the coordinate y used by Thomas ranges from -1 to 1 and the center of the channel is at $y = 0$. The coordinate y used herein ranges from 0 to 2, and the center of the channel is at $y = 1$. Also there is a difference in the definition of the stream function used by Thomas and the definition used herein. When these factors are considered, it can be seen that there is good agreement between the values taken on by the eigenfunction reported herein and the values given by Thomas. Although this agreement cannot be seen readily for all the values because of the difference in the increment of the independent variable y, there are several values that can be checked directly. These values agree with the results of Thomas shown in table II to four decimal places.

TABLE II. - EIGENFUNCTIONS FOR WAVE NUMBER OF 1,

REYNOLDS NUMBER OF 10,000

y	Present method (256 steps)		Method of Thomas	
	φ_r	φ_i	φ_r	φ_i
0	0	0	0	0
.0625	.083523	-.001321	------	------
.1250	.223679	-.013405	------	------
.1875	.359409	-.004517	------	------
.2500	.473718	-.003641	.473721	-.003630
.3125	.569594	-.003240	------	------
.3750	.652213	-.002655	------	------
.4375	.723638	-.002131	------	------
.5000	.785187	-.001668	.785190	-.001662
.5625	.837814	-.001266	------	------
.6250	.882240	-.000923	------	------
.6875	.919018	-.000637	------	------
.7500	.948579	-.000406	.948579	-.000404
.8125	.971250	-.000227	------	------
.8750	.987277	-.000101	------	------
.9375	.996827	-.000025	------	------
1.0000	1.000000	0	1.000000	0

Finally, the value of the minimum critical Reynolds number was obtained. The eigenvalues were obtained in the vicinity of the minimum value of 5780 reported in reference 4. Figure 1(a) shows c_i plotted against α for various values of Re. Interpolation (fig. 1(b)) based on the values given in figure 1(a) leads to a minimum critical Reynolds number (the lowest value of Re for which instability exists) of 5767 at $\alpha = 1.02$. An interpolated value of 5780 at $\alpha = 1.026$ is given in reference 4.

From the comparisons made previously, it can be seen that the agreement of the results of this report with those of Thomas is very good.

CONCLUDING REMARKS

The calculations and a comparison of them with the method of Thomas indicate that the method reported herein is rapid and highly accurate. Less than 1 minute of computing time on the IBM 7094 computer is required to calculate the eigenvalues at a representative point in the wave number-Reynolds number diagram if reasonably accurate initial estimates of the eigenvalues are provided. The method appears capable of being easily generalized to solve more complex stability problems. The close agreement of the results presented in this report with the results of Thomas is gratifying in view of the previous history of contradictory results regarding the stability of plane Poiseuille flow.

Lewis Research Center
National Aeronautics and Space Administration
Cleveland, Ohio, May 19, 1964

APPENDIX A

SYMBOLS

c	phase velocity
L	channel half-spacing
p	$s(0)$
q	$s'(0)$
Re	Reynolds number
r	$s(1)$
s	disturbance vorticity amplitude
t	time
U	velocity of basic flow
u	disturbance velocity parallel to plates
v	disturbance velocity normal to plates
x	distance parallel to plates
y	normal distance from lower plate
y_c	matching point
α	wave number
ν	kinematic viscosity
φ	stream function amplitude
ψ	stream function

Subscripts:

b	refers to backwards solution
f	refers to forward solution
i	refers to imaginary part
max	maximum

r refers to real part

, denotes partial differentiation

Superscripts:

$(\bar{\ })$ dimensional quantity

' denotes differentiation with respect to y

APPENDIX B

EQUATIONS IN REAL FORM

The real differential equations are obtained by separating the original equations into real and imaginary parts. For example, equations (8) written in real form are as follows:

$$\varphi_r'' = s_r + \alpha^2\varphi_r \tag{B1a}$$

$$\varphi_i'' = s_i + \alpha^2\varphi_i \tag{B1b}$$

$$s_r'' = \alpha^2 s_r - \alpha Re\left[(2y - y^2 - c_r)s_i - c_i s_r + 2\varphi_i\right] \tag{B1c}$$

$$s_i'' = \alpha^2 s_i + \alpha Re\left[(2y - y^2 - c_r)s_r + c_i s_i + 2\varphi_r\right] \tag{B1d}$$

The perturbation differential equations for variations with respect to the real part of the initial values for both the forward and backward solutions are of the same form as equations (B1).

The perturbation differential equations for the variation with respect to the eigenvalue c_r for both the forward and backward solutions are of the following form:

$$\varphi_{r,c_r}'' = s_{r,c_r} + \alpha^2\varphi_{r,c_r} \tag{B2a}$$

$$\varphi_{i,c_r}'' = s_{i,c_r} + \alpha^2\varphi_{i,c_r} \tag{B2b}$$

$$s_{r,c_r}'' = \alpha^2 s_{r,c_r} - \alpha Re\left[(2y - y^2 - c_r)s_{i,c_r} - c_i s_{r,c_r} + 2\varphi_{i,c_r} - s_i\right] \tag{B2c}$$

$$s_{i,c_r}'' = \alpha^2 s_{i,c_r} + \alpha Re\left[(2y - y^2 - c_r)s_{r,c_r} + c_i s_{i,c_r} + 2\varphi_{r,c_r} - s_r\right] \tag{B2d}$$

The real linear equations for the corrections to the initial conditions, and the eigenvalue are obtained by separating the original equations into real and imaginary parts. For example, equation (21a) written in real form leads to the two real equations

$$\begin{aligned}(\varphi_f)_r - (\varphi_b)_r + (\varphi_{f,p})_r\,\Delta p_r - (\varphi_{f,p})_i\,\Delta p_i + (\varphi_{f,q})_r\,\Delta q_r - (\varphi_{f,q})_i\,\Delta q_i \\ - (\varphi_{b,r})_r\,\Delta r_r + (\varphi_{b,r})_i\,\Delta r_i + \left[(\varphi_{f,c})_r - (\varphi_{b,c})_r\right]\Delta c_r \\ - \left[(\varphi_{f,c})_i - (\varphi_{b,c})_i\right]\Delta c_i = 0\end{aligned} \tag{B3a}$$

$$(\varphi_f)_i - (\varphi_b)_i + (\varphi_{f,p})_i \Delta p_r + (\varphi_{f,p})_r \Delta p_i + (\varphi_{f,q})_i \Delta q_r + (\varphi_{f,q})_r \Delta q_i$$

$$- (\varphi_{b,r})_i \Delta r_r - (\varphi_{b,r})_r \Delta r_i + \left[(\varphi_{f,c})_i - (\varphi_{b,c})_i\right]\Delta c_r$$

$$- \left[(\varphi_{f,c})_r - (\varphi_{b,c})_r\right]\Delta c_i = 0 \qquad \text{(B3b)}$$

where

$$(\varphi_{f,q})_r = \frac{q_r(\varphi_f)_r + q_i(\varphi_f)_i - \left[q_r p_r + q_i p_i\right](\varphi_{f,p})_r - \left[q_i p_r - q_r p_i\right](\varphi_{f,p})_i}{q_r^2 + q_i^2} \qquad \text{(B4a)}$$

$$(\varphi_{f,q})_i = \frac{q_r(\varphi_f)_i - q_i(\varphi_f)_r - \left[q_r p_r + q_i p_i\right](\varphi_{f,p})_i + \left[q_i p_r - q_r p_i\right](\varphi_{f,p})_r}{q_r^2 + q_i^2} \qquad \text{(B4b)}$$

Since the derivatives with respect to the real quantities are the ones that are calculated, it is necessary to express the real and imaginary parts of the complex derivatives in terms of derivatives with respect to real quantities. For example, in the case of $\varphi_{f,p}$, $(\varphi_{f,p})_r = (\varphi_f)_{r,p_r}$ and $(\varphi_{f,p})_i = (\varphi_f)_{i,p_r}$.

APPENDIX C

INTEGRATION FORMULAS

The integration is performed by using the fifth-order predictor-corrector method of Milne, which uses the fourth-order Runge-Kutta method to obtain starting values.

Let the system of n equations to be solved be given in the form

$$y_i'' = f_i(x, y_1, y_2, \ldots, y_n), \qquad (i = 1, 2, \ldots, n) \tag{C1}$$

with the initial conditions

$$y_i(x_o) = y_{io}, \qquad y_i'(x_o) = y_{io}', \qquad (i = 1, 2, \ldots, n) \tag{C2}$$

Let $y_{i,k}$ and $y_{i,k}'$ be the values of y_i and y_i' at $x = x_k$, $f_{i,k}$ be the second derivative of y_i at $x = x_k$, and h be the step size. The special Runge-Kutta formulas (ref. 8) used are as follows:

$$k_{i1} = hf_i(x_k, y_{i,k}) \tag{C3a}$$

$$k_{i2} = hf_i\left(x_k + \frac{h}{2}, y_{ik} + \frac{h}{2} y_{i,k}' + \frac{h}{8} k_{i1}\right) \tag{C3b}$$

$$k_{i3} = hf_i\left(x_k + h, y_{ik} + hy_{ik}' + \frac{h}{2} k_{i2}\right) \tag{C3c}$$

$$y_{i,k+1} = y_{i,k} + h\left[y_{i,k}' + \frac{1}{6} (k_{i1} + 2k_{i2})\right] \tag{C3d}$$

$$y_{i,k+1}' = y_{i,k}' + \frac{h}{6} \left[k_{i1} + 4k_{i2} + k_{i3}\right] \tag{C3e}$$

where $f_i(x_k, y_{i,k})$ is a shorthand notation for $f_i(x_k, y_{1,k}, y_{2,k}, \ldots, y_{n,k})$.

The Milne predictor-corrector formulas (ref. 6) for solving the system (C1) are

$$p_{i,k+1} = y_{i,k} + y_{i,k-2} - y_{i,k-3} + \frac{h^2}{4} (5f_{i,k} + 2f_{i,k-1} + 5f_{i,k-2}) \tag{C4a}$$

$$y_{i,k+1} = 2y_{i,k} - y_{i,k-1} + \frac{h^2}{12} \left[f_i(x_{k+h}, p_{i,k+1}) + 10f_{i,k} + f_{i,k-1}\right] \tag{C4b}$$

The corrector formula equation (C4b) is applied only once so that only two derivative evaluations are needed for each Milne integration step. The starting values needed in the predictor formula (eq. (C4a)) are obtained by using equations (C3).

APPENDIX D

DESCRIPTION OF THE FORTRAN PROGRAM FOR SOLUTION OF THE EIGENVALUE PROBLEM OF PLANE POISEUILLE FLOW

The numerical procedure outlined previously for solving the eigenvalue problem was programed for solution on the IBM 7094 in FORTRAN IV. The program as listed below is available upon request from the author.

The correspondence between FORTRAN symbols used in this program and the mathematical notation employed previously is shown in the following list:

FORTRAN symbol	Mathematical symbol	FORTRAN symbol	Mathematical symbol
Y1	φ_r	DS2C	s_i', c_r
Y2	φ_i	C1	c_r
S1	s_r	C2	c_i
S2	s_i	DELA1	Δp_r
Y1A	φ_r, p_r	DELA2	Δp_i
Y2A	φ_i, p_r	DELB1	Δq_r
S1A	s_r, p_r	DELB2	Δq_i
S2A	s_i, p_r	DELC1	Δc_r
Y1C	φ_r, c_r	DELC2	Δc_i
Y2C	φ_i, c_r	DELD1	Δr_r
S1C	s_r, c_r	DELD2	Δr_i
S2C	s_i, c_r	S1FWD	$s_r(0)$
DY1	φ_r'	S2FWD	$s_i(0)$
DY2	φ_i'	DS1FWD	$s_r'(0)$
DS1	s_r'	DS2FWD	$s_i'(0)$
DS2	s_i'	S1BACK	$s_r(1)$
DY1A	φ_r', p_r	S2BACK	$s_i(1)$
DY2A	φ_i', p_r	A	α
DS1A	s_r', p_r	R	Re
DS2A	s_i', p_r	W	$2y - y^2$
DY1C	φ_r', c_r	DDW	-2
DY2C	φ_i', c_r	X	y
DS1C	s_r', c_r		

The following remarks are intended to aid in a study of the program:

(1) Subroutine DAUX is used to evaluate the second derivatives. The variables Z and DDZ that appear in DAUX are dummy variables.

(2) Subroutine ZMANDZ is used to store the matrix of coefficients that are formed from functions and partial derivatives evaluated at the matching point. The solution of the simultaneous linear equations is accomplished by calling subroutine LSGAUS. A listing of this subroutine is not included herein since programs that solve simultaneous linear equations are readily available at all computing establishments. For the purpose of following the logic of subroutine ZMANDZ, the reader can ignore all the arguments in the call of LSGAUS except EE and VV. Before the subroutine is called, EE contains the coefficient matrix and VV contains the "right-hand side." After LSGAUS is called, VV contains the answers.

(3) Subroutine INTEGR carries out the step-by-step integration with either the Runge-Kutta method (INDEX = 0) or the Milne method, which uses the Runge-Kutta method to obtain starting values (INDEX = 1).

The program listing is given in appendix E and flow charts of the program are presented in figures 2 to 4.

APPENDIX E

PROGRAM LISTING

```
      MAIN
      EXTERNAL DAUX
      DOUBLE PRECISION Y1,Y2,S1,S2,Y1A,Y2A,S1A,S2A,Y1C,Y2C,S1C,S2C,DY1,
     1DY2,DS1,DS2,DY1A,DY2A,DS1A,DS2A,DY1C,DY2C,DS1C,DS2C,C1,C2,T,DT,
     2DELA1,DELA2,DELB1,DELB2,DELC1,DELC2,DELD1,DELD2,S1FWD,S2FWD,
     3DS1FWD,DS2FWD,S1BACK,S2BACK
      DOUBLE PRECISION DDT,SMALLE,SMALLN
      COMMON C1,C2,A,R,W,DDW,AA,AR
      COMMON S1FWD,S2FWD,DS1FWD,DS2FWD,DELA1,DELA2,DELB1,DELB2,DELC1,
     1DELC2,DELD1,DELD2,T,DT
      DIMENSION T(12),DT(12),DDT(12)
      EQUIVALENCE (Y1,T(1)),(Y2,T(2)),(S1,T(3)),(S2,T(4)),(Y1A,T(5)),
     1(Y2A,T(6)),(S1A,T(7)),(S2A,T(8)),(Y1C,T(9)),(Y2C,T(10)),(S1C,T(11)
     2),(S2C,T(12)),(DY1,DT(1)),(DY2,DT(2)),(DS1,DT(3)),(DS2,DT(4)),
     3(DY1A,DT(5)),(DY2A,DT(6)),(DS1A,DT(7)),(DS2A,DT(8)),(DY1C,DT(9)),
     4(DY2C,DT(10)),(DS1C,DT(11)),(DS2C,DT(12))
  201 FORMAT(14I5)
    9 READ(5,201)INDEX,N,ITERAT
  202 FORMAT(7F10.0)
      READ(5,202)H,DELXPR,XEND,XMATCH
  204 FORMAT(1P4D20.13)
      READ(5,204)SMALLE,SMALLN
      READ(5,202)A,R
      AA=A**2
      AR=A*R
   30 READ(5,204)S1FWD,S2FWD,DS1FWD,DS2FWD,C1,C2,S1BACK,S2BACK
  101 FORMAT(7H1INDEX=I5,4H  N=I5,9H  ITERAT=I5)
   31 WRITE(6,101)INDEX,N,ITERAT
  102 FORMAT(3H H=1PE14.7,9H  DELXPR=1PE14.7,7H  XEND=1PE14.7,9H  XMATCH
     1=1PE14.7)
      WRITE(6,102)H,DELXPR,XEND,XMATCH
  118 FORMAT(8H SMALLE=1PD22.15,9H  SMALLN=1PD22.15)
      WRITE(6,118)SMALLE,SMALLN
  103 FORMAT(3H A=1PE14.7,4H  R=1PE14.7)
      WRITE(6,103)A,R
  104 FORMAT(7H S1FWD=1PD22.15,8H  S2FWD=1PD22.15,9H  DS1FWD=1PD22.15,9H
     1  DS2FWD=1PD22.15)
  105 FORMAT(4H C1=1PD22.15,5H  C2=1PD22.15,9H  S1BACK=1PD22.15,9H  S2BA
     1CK=1PD22.15)
      WRITE(6,104)S1FWD,S2FWD,DS1FWD,DS2FWD
      WRITE(6,105)C1,C2,S1BACK,S2BACK
   10 I=1
      J=1
   49 CONTINUE
      Y1A=.0D0
      Y2A=.0D0
      DY1A=.0D0
      DY2A=.0D0
      S1A=1.D0
      S2A=.0D0
      DS1A=.0D0
      DS2A=.0D0
      Y1C=.0D0
      Y2C=.0D0
      DY1C=.0D0
      DY2C=.0D0
      S1C=.0D0
      S2C=.0D0
      DS1C=.0D0
      DS2C=.0D0
      GO TO(50,51),J
```

```
 50 J = 2
    X = .0
    XPRINT=.0
    DELXPR = ABS(DELXPR)
    H=ABS(H)
    Y1=.0D0
    Y2=.0D0
    DY1=.0D0
    DY2=.0D0
    S1=S1FWD
    S2=S2FWD
    DS1=DS1FWD
    DS2=DS2FWD
    GO TO 4
 51 J = 1
    X = XEND
    H=-ABS(H)
    XPRINT = XEND
    DELXPR=-ABS(DELXPR)
    Y1=1.D0
    Y2=.0D0
    DY1=.0D0
    DY2=.0D0
    S1=S1BACK
    S2=S2BACK
    DS1=.0D0
    DS2=.0D0
106 FORMAT(110HOX Y1 Y2 DY1 DY2 S1 S2 DS1 DS2/ W Y1A Y2A DY1A DY2A S1A
   1 S2A DS1A DS2A/ DDW Y1C Y2C DY1C DY2C S1C S2C DS1C DS2C)
  4 WRITE (6,106)
  6 CALL INTEGR (N,H,X,0,T,DT,DDT,INDEX,DAUX)
    GO TO 14
 15 CALL INTEGR(N,H,X,1,T,DT,DDT,INDEX,DAUX)
    GO TO (61,60),J
 60 XX=X-XPRINT
 64 IF(XX)15,14,14
107 FORMAT(F14.4,1P8D14.5/(1PE14.5,1P8D14.5))
 14 WRITE(6,107)X,Y1,Y2,DY1,DY2,S1,S2,DS1,DS2,W,Y1A,Y2A,DY1A,DY2A,S1A,
   1S2A,DS1A,DS2A,DDW,Y1C,Y2C,DY1C,DY2C,S1C,S2C,DS1C,DS2(
  3 XPRINT=XPRINT+DELXPR
    GO TO (63,62),J
 62 XXX = X-XMATCH
 65 IF (XXX) 15,16,16
 61 XX = XPRINT-X
    GO TO 64
 63 XXX = XMATCH-X
    GO TO 65
 16 CONTINUE
    CALL ZMANDZ (I)
    IF(I)57,57,58
 58 I=I-1
    GO TO 49
110 FORMAT(16HOSUM OF SQUARES=1PD14.7,18H  SUM OF EIGEN SQ=1PD14.7)
 57 WRITE(6,110)Y1,Y2
    S1FWD=S1FWD+DELA1
    S2FWD=S2FWD+DELA2
    DS1FWD=DS1FWD+DELB.
    DS2FWD=DS2FWD+DELB2
    C1=C1+DELC1
    C2=C2+DELC2
    S1BACK=S1BACK+DELD1
    S2BACK=S2BACK+DELD2
    WRITE(6,104)S1FWD,S2FWD,DS1FWD,DS2FWD
    WRITE(6,105)C1,C2,S1BACK,S2BACK
    ITERAT=ITERAT-1
    IF(ITERAT)9,9,28
 28 IF(Y1-SMALLN)9,9,29
 29 IF(Y2-SMALLE)9,9,10
    END
```

```
      SUBROUTINE INTEGR(N,H,X,ISET,Y,DY,DDY,INDEX,F)
      DOUBLE PRECISION E,YLLL,YLL,YL,Y,DYL,DY,DDYLL,DDYL,DDY,YR,DYR,DDYR
     1,C2,C3,P
      DIMENSION YLLL(12),YLL(12),YL(12),Y(12),DYL(12),DY(12),DDYLL(12),
     1DDYL(12),DDY(12),YR(12),DYR(12),DDYR(12),C2(12),C3(12),P(12)
      E=H
      IF(ISET)6,6,7
    6 IF(INDEX)9,9,8
    8 ASSIGN 2 TO K
      GO TO 21
    9 ASSIGN 1 TO K
   21 CALL F(X,Y,DDY)
      GO TO 10
    7 GO TO K,(1,2,3,4,5)
    1 DO 11  I=1,N
   11 P(I) = Y(I)+(H/2.)*DY(I)+((H*H)/8.)*DDY(I)
      CALL F(X+H/2.,P,C2)
      DO 12  I=1,N
   12 P(I) = Y(I)+H*DY(I)+((H*H)/2.)*C2(I)
      CALL F(X+H    ,P,C3)
      DO 13  I=1,N
      YR(I) = Y(I)+H*(DY(I)+(E/6.)*(DDY(I)+2.*C2(I)))
   13 DYR(I) = DY(I) + (E/6.)*(DDY(I)+4.*C2(I)+C3(I))
      CALL F(X+H,YR,DDYR)
   22 X = X+H
      DO 14  I=1,N
      YLLL(I) = YLL(I)
      YLL(I) = YL(I)
      YL(I) = Y(I)
      Y(I) = YR(I)
      DYL(I) = DY(I)
      DY(I) = DYR(I)
      DDYLL(I) = DDYL(I)
      DDYL(I) = DDY(I)
   14 DDY(I) = DDYR(I)
   10 RETURN
    2 ASSIGN 3 TO K
      GO TO 1
    3 ASSIGN 4 TO K
      GO TO 1
    4 ASSIGN 5 TO K
      GO TO 1
    5 DO 15  I=1,N
   15 P(I) = Y(I)+YLL(I)-YLLL(I)+((H*H)/4.)*(5.*DDY(I)+2.*DDYL(I)+5.*DDY
     1LL(I))
      CALL F(X+H,P,DDYR)
      DO 16  I=1,N
   16 YR(I)=2.*Y(I)-YL(I)+((E*E)/12.)*(DDYR(I)+10.*DDY(I)+DDYL(I))
      CALL F(X+H,YR,DDYR)
      DO 17  I=1,N
   17 DYR(I) = DYL(I) +(E/3.)*(DDYR(I)+4.*DDY(I)+DDYL(I))
      GO TO 22
      END

      SUBROUTINE DAUX (X,Z,DDZ)
      DOUBLE PRECISION C1,C2,Z,DDZ
      COMMON C1,C2,A,P,W,DDW,AA,AR
      DIMENSION Z(12),DDZ(12)
      W=2.*X-X*X
      DDW=-2.
      DDZ(1)= AA*Z(1) +Z(3)
      DDZ(2)= AA*Z(2) +Z(4)
      DDZ(3)= AA*Z(3) -AR*((W-C1)*Z(4) -C2*Z(3) -DDW*Z(2))
      DDZ(4)= AA*Z(4) +AR*((W-C1)*Z(3) +C2*Z(4) -DDW*Z(1))
      DDZ(5)= AA*Z(5) +Z(7)
      DDZ(6)= AA*Z(6) +Z(8)
      DDZ(7)= AA*Z(7) -AR*((W-C1)*Z(8) -C2*Z(7) -DDW*Z(6))
      DDZ(8)= AA*Z(8) +AR*((W-C1)*Z(7) +C2*Z(8) -DDW*Z(5))
      DDZ(9)= AA*Z(9) +Z(11)
      DDZ(10)=AA*Z(10)+Z(12)
      DDZ(11)=AA*Z(11)-AR*((W-C1)*Z(12)-C2*Z(11)-DDW*Z(10)-Z(4))
      DDZ(12)=AA*Z(12)+AR*((W-C1)*Z(11)+C2*Z(12)-DDW*Z(9)-Z(3))
      RETURN
      END
```

```
      SUBROUTINE ZMANDZ (I)
      DOUBLE PRECISION Y1,Y2,S1,S2,Y1A,Y2A,S1A,S2A,Y1C,Y2C,S1C,S2C,DY1,
     1DY2,DS1,DS2,DY1A,DY2A,DS1A,DS2A,DY1C,DY2C,DS1C,DS2C,C1,C2,T,DT,
     2DELA1,DELA2,DELB1,DELB2,DELC1,DELC2,DELD1,DELD2,S1FWD,S2FWD,
     3DS1FWD,DS2FWD,S1BACK,S2BACK
      DOUBLE PRECISION EF,VV
      COMMON C1,C2,A,B,W,DDW,AA,AB
      COMMON S1FWD,S2FWD,DS1FWD,DS2FWD,DELA1,DELA2,DELB1,DELB2,DELC1,
     1DELC2,DELD1,DELD2,T,DT
      DIMENSION T(12),DT(12),EF(8,8),VV(8)
      EQUIVALENCE (Y1,T(1)),(Y2,T(2)),(S1,T(3)),(S2,T(4)),(Y1A,T(5)),
     1(Y2A,T(6)),(S1A,T(7)),(S2A,T(8)),(Y1C,T(9)),(Y2C,T(10)),(S1C,T(11)
     2),(S2C,T(12)),(DY1,DT(1)),(DY2,DT(2)),(DS1,DT(3)),(DS2,DT(4)),
     3(DY1A,DT(5)),(DY2A,DT(6)),(DS1A,DT(7)),(DS2A,DT(8)),(DY1C,DT(9)),
     4(DY2C,DT(10)),(DS1C,DT(11)),(DS2C,DT(12))
      IF(I) 52,52,53
C     FORWARD 53
   53 VV(1)=Y1
      VV(2)=Y2
      VV(3)=DY1
      VV(4)=DY2
      VV(5)=S1
      VV(6)=S2
      VV(7)=DS1
      VV(8)=DS2
      EE(1,1)=Y1A
      EE(2,1)=Y2A
      EE(3,1)=DY1A
      EE(4,1)=DY2A
      EE(5,1)=S1A
      EE(6,1)=S2A
      EE(7,1)=DS1A
      EE(8,1)=DS2A
      DENOM=DS1FWD**2+DS2FWD**2
      A1=DS1FWD/DENOM
      A2=DS2FWD/DENOM
      AA1=(S1FWD*DS1FWD+S2FWD*DS2FWD)/DENOM
      AA2=(S1FWD*DS2FWD-S2FWD*DS1FWD)/DENOM
      EE(1,3)=A1*Y1+A2*Y2-AA1*Y1A-AA2*Y2A
      EE(2,3)=A1*Y2-A2*Y1-AA1*Y2A+AA2*Y1A
      EE(3,3)=A1*DY1+A2*DY2-AA1*DY1A-AA2*DY2A
      EE(4,3)=A1*DY2-A2*DY1-AA1*DY2A+AA2*DY1A
      EE(5,3)=A1*S1+A2*S2-AA1*S1A-AA2*S2A
      EE(6,3)=A1*S2-A2*S1-AA1*S2A+AA2*S1A
      EE(7,3)=A1*DS1+A2*DS2-AA1*DS1A-AA2*DS2A
      EE(8,3)=A1*DS2-A2*DS1-AA1*DS2A+AA2*DS1A
      EE(1,5)=Y1C
      EE(2,5)=Y2C
      EE(3,5)=DY1C
      EE(4,5)=DY2C
      EE(5,5)=S1C
      EE(6,5)=S2C
      EE(7,5)=DS1C
      EE(8,5)=DS2C
      GO TO 56
      BACKWARD 52
   52 VV(1)=Y1-VV(1)
      VV(2)= Y2-VV(2)
      VV(3)=DY1-VV(3)
      VV(4)=DY2-VV(4)
      VV(5)=S1-VV(5)
```

```
      VV(6)=S2-VV(6)
      VV(7)=DS1-VV(7)
      VV(8)=DS2-VV(8)
      EE(1,5)=EE(1,5)-Y1C
      EE(2,5)=EE(2,5)-Y2C
      EE(3,5)=EE(3,5)-DY1C
      EE(4,5)=EE(4,5)-DY2C
      EE(5,5)=EE(5,5)-S1C
      EE(6,5)=EE(6,5)-S2C
      EE(7,5)=EE(7,5)-DS1C
      EE(8,5)=EE(8,5)-DS2C
      EE(1,7)=-Y1A
      EE(2,7)=-Y2A
      EE(3,7)=-DY1A
      EE(4,7)=-DY2A
      EE(5,7)=-S1A
      EE(6,7)=-S2A
      EE(7,7)=-DS1A
      EE(8,7)=-DS2A
C     EVEN COLUMNS
      DO 100 K=1,4
      DO 100 L=1,4
      EE(2*L-1,2*K)=-EE(2*L,2*K-1)
      EE(2*L  ,2*K)= EE(2*L-1,2*K-1)
  100 CONTINUE
      Y1=.0D0
      DO 6 L=1,8
    6 Y1=Y1+VV(L)*VV(L)
    8 CALL LSGAUS(EE,VV,8,8,.0D0,IYESNO)
      Y2=.0D0
      DO 12 L=1,8
   12 Y2=Y2+VV(L)*VV(L)
    4 DELA1=VV(1)
      DELA2=VV(2)
      DELB1=VV(3)
      DELB2=VV(4)
      DELC1=VV(5)
      DELC2=VV(6)
      DELD1=VV(7)
      DELD2=VV(8)
      GO TO 56
   56 RETURN
      END
```

REFERENCES

1. Heisenberg, W.: On Stability and Turbulence of Fluid Flows. (Über Stabilität und Turbulenz von Flüssigkeitsströmen.) NACA TM 1291, 1951. (Trans. from Annalen der Phys., Bd. 74, no. 15, 1924, pp. 577-627.)

2. Lin, C. C.: On the Stability of Two-Dimensional Parallel Flows, pt. I. Quarterly Appl. Math., vol. III, July 1945, pp. 117-142; pt. II, vol. III, Oct. 1945, pp. 218-234; pt. III, vol. III, Jan. 1946, pp. 277-301.

3. Pekeris, C. L.: Stability of the Laminar Parabolic Flow of a Viscous Fluid Between Parallel Fixed Walls. Phys. Rev., no. 2, vol. 74, July 15, 1948, pp. 191-199.

4. Thomas, L. H.: The Stability of Plane Poiseuille Flow. Phys. Rev., vol. 91, no. 4, Aug. 15, 1953, pp. 780-783.

5. Lin, C. C.: Theory of Hydrodynamic Stability. Cambridge Univ. Press, 1955.

6. Hildebrand, F. B.: Introduction to Numerical Analysis. McGraw-Hill Book Co., Inc., 1956, pp. 223-226.

7. Fox, Leslie: Some Numerical Experiments with Eigenvalue Problems in Ordinary Differential Equations. Boundary Problems in Differential Equations, R. E. Langer, ed., Univ. of Wisconsin Press, 1960, pp. 243-255.

8. Scarborough, J. B.: Numerical Mathematical Analysis. Fourth ed., Johns Hopkins Press, 1958, pp. 316-317.



(a) Eigenvalues.



(b) Linear interpolation.

Figure 1. - Minimum critical Reynolds number.



Figure 2. - Flow chart of main program.



Figure 3. - Flow chart of INTEGR subroutine.



Figure 4. - Flow chart of ZMANDZ subroutine.

NASA-Langley, 1964 E-2449

2/7/85

"The aeronautical and space activities of the United States shall be conducted so as to contribute . . . to the expansion of human knowledge of phenomena in the atmosphere and space. The Administration shall provide for the widest practicable and appropriate dissemination of information concerning its activities and the results thereof."

—NATIONAL AERONAUTICS AND SPACE ACT OF 1958

NASA SCIENTIFIC AND TECHNICAL PUBLICATIONS

TECHNICAL REPORTS: Scientific and technical information considered important, complete, and a lasting contribution to existing knowledge.

TECHNICAL NOTES: Information less broad in scope but nevertheless of importance as a contribution to existing knowledge.

TECHNICAL MEMORANDUMS: Information receiving limited distribution because of preliminary data, security classification, or other reasons.

CONTRACTOR REPORTS: Technical information generated in connection with a NASA contract or grant and released under NASA auspices.

TECHNICAL TRANSLATIONS: Information published in a foreign language considered to merit NASA distribution in English.

TECHNICAL REPRINTS: Information derived from NASA activities and initially published in the form of journal articles.

SPECIAL PUBLICATIONS: Information derived from or of value to NASA activities but not necessarily reporting the results of individual NASA-programmed scientific efforts. Publications include conference proceedings, monographs, data compilations, handbooks, sourcebooks, and special bibliographies.

Details on the availability of these publications may be obtained from:

SCIENTIFIC AND TECHNICAL INFORMATION DIVISION

NATIONAL AERONAUTICS AND SPACE ADMINISTRATION

Washington, D.C. 20546